

实用定量管理程序集

上海工程技术大学 徐志良

SHIYONG DINGLIANG GUANLI CHENGXUJI

清华大学出版社

实用定量管理程序集

上海工程技术大学

徐志良 编著

清华大学出版社

内 容 提 要

本书把工厂企业中常用的定量管理技术方法(运筹学的各个重要分枝)编成相应的通用程序,共计十二个。全部程序均采用BASIC语言编写,适用于各种不同型号的微型计算机。

书中每个程序都配有数学模型、算法原理、使用说明以及各种典型实例的运用,方便读者学习。

本书可作为高等院校管理专业的本科生和研究生的辅助教材和参考读物,也可作为各类管理干部培训班的运筹学课程的教学演示材料。同时,也可作为工厂企业的管理人员和其他工程技术人员从事计算机管理应用的实用手册。

实用定量管理程序集

上海工程技术大学

徐志良 编著

☆

清华大学出版社出版

北京 清华园

北京京辉印刷厂印刷

新华书店北京发行所发行

☆

开本: 787×1092 1/16 印张: 17.25 字数: 43.8千字

1987年5月第1版 1987年5月第1次印刷

印数: 0001-5000

统一书号: 15235·273 定价: 3.55元

前 言

现代化企业管理就是管理科学加计算机。随着经济建设的迅速发展,企业内部之间以及企业与外界社会之间的信息交流日趋频繁。为了促进生产,发展经济,我们必须采用现代化企业管理。为此,计算机在工厂企业管理中的应用是一个十分重要的课题。

近年来,全国许多工厂企业已纷纷购置了计算机(其中大部分为微机),并正在进行管理信息系统的开发。计算机已开始应用于工厂企业的一些职能部门,比如财务会计报表,累计合同台帐,月销售快报,市场分析表,利润预测报告等等。但必须指出,从普遍情况来看,当前我国一般企业的信息管理系统的开发还处于低层次的阶段,也就是说还停留在模拟人工管理系统的阶段,即用计算机代替人工进行一些数据处理的工作,从而节省一些人力并缩短处理时间。对于现代化企业管理来说,仅仅做到这一步,还是远远不够的。我们必须在数据处理的基础上,运用各种定量管理技术,建立一系列完整的数学模型,借助计算机进行优化求解,从大量的数据信息中求出最优的解值,为企业领导的经营决策和生产管理提供更正确的定量依据。这是以前人工管理系统远不能做到,也是目前管理信息系统初级的数据处理阶段尚未能达到的。由此可见,管理信息系统从单纯数据处理的低层次向优化求解的高层次的过渡,是非常必要的。

在这一想法的促使下,我们上海工程技术大学管理工程系从1983年开始,历时约二年,把定量管理技术(运筹学)的各个重要分枝(尤其是企业管理中实用的)提炼成具体的算法,然后设计编制成通用程序,并在MC 68000微机上全部调试通过,以供广大微机用户使用。

本书有以下几个特点:

1. 书中的全部程序均采用 BASIC 语言编写,力求简单易学,便于推广应用,且一般微机均兼容(如IBM-PC等)。
2. 努力使各程序的算法严密,有所创新,尽量采用多种计算技巧,从而达到运算速度快,能求解大型的实际问题。
3. 力求做到通用性强。这表现在:首先充分考虑了数学模型的全面性、逻辑性,使这些程序能适应各种实际问题,不会因为情况稍有变化而失效。其次,这些程序是按正规严密的理论算法编制而成,既可作为管理科研的现存计算工具,又可作为各种管理专业教学的实验示例手段。
4. 力求做到具有一定的灵活性和独立性。这些程序既可通过人机对话,用户能方便地独立运用;又可稍加修改,把这些程序变成一个功能模块,通过“接口”,TXT文件,和数据库语言dBASE-Ⅱ等联机使用。
5. 每一个通用程序均配有数学模型、算法原理、使用说明以及各种典型实例的运用,以使用户、读者能随意查找,无师自通。

本书的两点说明:

1. 为了验证算法的正确性,本书各程序的上机运行实例尽量选取其他书本上的现存例

子。

2. 本书各程序的数组变量容量定维是按照 MC68000 和 IBM-PC 的数组最大可容元素 32767 而设计的,如用户要把这些程序移植到其他较小内存容量的微机(如 Apple-Ⅱ 等)上,则可以对各程序的 DIM 语句中的数组定维缩小。

在本书出版之际,谨向长期对作者悉心指导的 Robert E. Machol 教授(美国西北大学,系统分析专家)、John J. Bartholdi 教授(美国佐治亚工学院,运筹学专家)、Stephen C. Graves 教授(美国麻省理工学院,管理科学专家)表示衷心的感谢。

上海交通大学周志诚教授在本程序集的研制和本书的编著期间,给予了热情的鼓励和极大的支持,并审阅了程序集,提出了许多宝贵的建议。我校管理系蒋仲刚同志认真地审阅了全书,作了仔细的校对和修改。在本程序集的编制过程中得到了系领导的支持和不少同志的帮助,在此一并向他们表示衷心的感谢。

编制设计通用的定量管理程序集,是一项艰巨的工作。由于水平有限,经验不足,本书一定会存在不少疏漏谬误,请读者、用户提出宝贵的意见和建议。

编著者

1985年6月

目 录

前言

第一章 带敏感性分析的单纯形法线性规划程序	1
§1.1 程序 LPSIM	1
§1.2 线性规划问题及其数学模型.....	9
§1.3 线性规划的图解法.....	12
§1.4 用单纯形法求最大值问题.....	14
§1.5 最小值问题与其他类型的约束条件.....	19
§1.6 敏感性分析.....	20
§1.7 程序 LPSIM 的算法和功能.....	23
§1.8 程序 LPSIM 的使用方法.....	25
§1.9 典型例子的运行.....	26
第二章 改进单纯形法线性规划程序	34
§2.1 程序 RESIM	34
§2.2 改进单纯形法.....	39
§2.3 程序 RESIM 的算法和功能.....	45
§2.4 程序 RESIM 的使用方法.....	46
§2.5 典型例子的运行.....	47
第三章 分枝定界法整数规划程序	54
§3.1 程序 ILPBB	54
§3.2 整数规划问题的提出.....	65
§3.3 分枝定界法.....	66
§3.4 程序 ILPBB 的算法和功能.....	72
§3.5 程序 ILPBB 的使用方法.....	74
§3.6 典型例子的运行.....	74
第四章 隐枚举法 0-1 整数规划程序	80
§4.1 程序 D01 ILP	80
§4.2 0-1 整数规划的隐枚举法.....	86
§4.3 程序 D01 ILP 的算法和功能.....	90
§4.4 程序 D01 ILP 的使用方法.....	91
§4.5 典型例子的运行.....	92
第五章 运输问题 VAM 程序	99
§5.1 程序 TRAVAM.....	99
§5.2 运输问题的提出及其线性规划模型	104

§5.3	运输问题的 VAM 法	106
§5.4	程序 TRAVAM 的算法和功能	110
§5.5	程序 TRAVAM 的使用方法	111
§5.6	典型例子的运行	111
第六章	运输问题单纯形法(MODI)程序	120
§6.1	程序 TRASIM	120
§6.2	运输单纯形法(MODI)	128
§6.3	运输问题的一些特殊情况	133
§6.4	程序 TRASIM 的算法和功能	134
§6.5	程序 TRASIM 的使用方法	135
§6.6	典型例子的运行	135
第七章	分配问题匈牙利法程序	146
§7.1	程序 HUNGAR	146
§7.2	分配问题的提出及其线性规划模型	154
§7.3	匈牙利法	156
§7.4	程序 HUNGAR 的算法和功能	160
§7.5	程序 HUNGAR 的使用方法	161
§7.6	典型例子的运行	162
第八章	单目标规划程序	172
§8.1	程序 SWGOAL	172
§8.2	目标规划简介	176
§8.3	单目标规划的数学模型及其求解方法	177
§8.4	程序 SWGOAL 的算法和功能	180
§8.5	程序 SWGOAL 的使用方法	182
§8.6	典型例子的运行	182
第九章	多目标规划程序	187
§9.1	程序 MUGOAL	187
§9.2	多目标规划的数学模型及其求解方法	193
§9.3	程序 MUGOAL 的算法和功能	202
§9.4	程序 MUGOAL 的使用方法	203
§9.5	典型例子的运行	204
第十章	网络计划程序	212
§10.1	程序 PERT	213
§10.2	网络计划技术简介	217
§10.3	网络图的绘制	218
§10.4	网络时间的计算	220
§10.5	程序 PERT 的算法和功能	225
§10.6	程序 PERT 的使用方法	226
§10.7	典型例子的运行	226

第十一章 单队单服务台随机服务系统模拟程序	233
§11.1 程序 SINQUE	233
§11.2 引言	237
§11.3 排队论的基本概念	238
§11.4 从一个实例来研究排队问题	241
§11.5 排队分析中所使用的概率分布	242
§11.6 单通道泊松/负指数排队模型(M/M/1)	246
§11.7 经济分析	247
§11.8 排队的模拟	249
§11.9 程序SINQUE 的算法和功能	250
§11.10 程序SINQUE 的使用方法	252
§11.11 典型例子的运行	252
第十二章 单队多服务台随机服务系统模拟程序	255
§12.1 程序MLTQUE	255
§12.2 多通道泊松/负指数排队模型(M/M/K)	261
§12.3 其他排队模型	262
§12.4 程序MLTQUE 的算法和功能	263
§12.5 程序MLTQUE 的使用方法	264
§12.6 典型例子的运行	264

第一章 带敏感性分析的单纯形法线性规划程序

§1.1 程序LPSIM

```
0010 REM * * * * *
0020 REM *   INPUT SEGMENT *
0030 REM * * * * *
0040 REM N-----MAXIMUM / MINIMUM CODE (0/1)
0050 REM M-----NUMBER OF TOTAL CONSTRAINTS
0060 REM G-----NUMBER OF  $\geq$  CONSTRAINTS AMONG THEM
0070 REM W-----NUMBER OF DECISION VARIABLES
0080 PRINT "MAX PROBLEM (0) OR MIN PROBLEM (1) ? "
0090 INPUT X
0100 PRINT X
0110 PRINT "HOW MANY CONSTRAINTS ARE THERE TOTALLY? "
0120 INPUT M
0130 PRINT M
0140 PRINT "HOW MANY  $\geq$  CONSTRAINTS AMONG THEM? "
0150 INPUT G
0160 PRINT G
0170 PRINT "HOW MANY DECISION-VARIABLES ARE THERE? "
0180 INPUT W
0190 PRINT W
0200 N = W + G
0210 PRINT
0220 IF X = 0 THEN 250
0230 PRINT "THIS IS A MINIMUM PROBLEM"
0240 GOTO 260
0250 PRINT "THIS IS A MAXIMUM PROBLEM"
0260 PRINT
0270 REM * * * * *
0280 REM *   READ COEFFICIENT MATRIX *
0290 REM * * * * *
0300 DIM A (100,200)
0310 FOR I = 1 TO M + 1
```

```

0320 FOR J=1 TO M+N+3
0330 READ A(I,J)
0340 NEXT J
0350 NEXT I
0360 DIM O(100)
0370 FOR I=1 TO M
0380 O(I)=A(I+1,M+N+3)
0390 NEXT I
0400 A(M+2,1)=0
0410 A(M+2,2)=0
0420 REM *****
0430 REM *      FIND    C-Z    AND    Z      *
0440 REM *****
0450 FOR J=3 TO M+N+3
0460 S=0
0470 FOR I=2 TO M+1
0480 S=S+A(I,2)*A(I,J)
0490 NEXT I
0500 A(M+2,J)=A(1,J)-S
0510 NEXT J
0520 A(M+2,M+N+3)=-A(M+2,M+N+3)
0530 T=0
0540 PRINT"# OF CYCLE      CURRENT VALUE"
0545 PRINT"-----"
0550 REM *****
0560 REM *      CYCLES BEGIN      *
0570 REM *****
0580 REM
0590 REM *****
0600 REM *      FIND THE PIVOT COLUMN      *
0610 REM *****
0620 M9=0
0630 FOR J=3 TO M+N+2
0640 IF A(M+2,J)<=M9 THEN 670
0650 M9=A(M+2,J)
0660 K=J
0670 NEXT J
0680 REM *****
0690 REM * DETERMINE WHETHER IT APPROACHES OPTIMUM *

```

```

0700 REM * * * * *
0710 IF T<>0 THEN 740
0720 IF M9<=0 THEN 2720
0730 GOTO 780
0740 IF M9<=0 THEN 1230
0750 REM * * * * *
0760 REM *          FIND THE PIVOT ROW          *
0770 REM * * * * *
0780 DIM C(100)
0790 L1=1E+14
0800 FOR I=2 TO M+1
0810 IF A(I,K)<=0 THEN 860
0820 C(I)=A(I,M+N+3)/A(I,K)
0830 IF C(I)>=L1 THEN 860
0840 L1=C(I)
0850 P=I
0860 NEXT I
0870 IF L1=1E+14 THEN 2740
0880 REM * * * * *
0890 REM *          PIVOT  CALCULATION          *
0900 REM * * * * *
0910 H=A(P,K)
0920 FOR J=3 TO M+N+3
0930 A(P,J)=A(P,J)/H
0940 NEXT J
0950 FOR I=2 TO M+1
0960 IF I=P THEN 1010
0970 H=A(I,K)
0980 FOR J=3 TO M+N+3
0990 A(I,J)=A(I,J)-A(P,J)*H
1000 NEXT J
1010 NEXT I
1020 REM * * * * *
1030 REM *          ENTER THE NEW BASIC VARIABLE      *
1040 REM * * * * *
1050 A(P,1)=K-2
1060 A(P,2)=A(I,K)
1070 REM * * * * *
1080 REM *          FIND  C-Z  AND  Z          *

```

```

1090 REM * * * * *
1100 FOR J=3 TO M+N+3
1110 S=0
1120 FOR I=2 TO M+1
1130 S=S+A(I,2)*A(I,J)
1140 NEXT I
1150 A(M+2,J)=A(I,J)-S
1160 NEXT J
1170 A(M+2,M+N+3)=-A(M+2,M+N+3)
1180 T=T+1
1190 PRINT T,
1200 PRINT "Z=";
1210 PRINT A(M+2,M+N+3)
1220 GOTO 590
1230 REM * * * * *
1240 REM *      OUTPUT SOLUTION AND OPTIMAL Z      *
1250 REM * * * * *
1251 DIM B(200)
1252 FOR J=1 TO M+N
1253 B(J)=0
1254 NEXT J
1255 FOR I=2 TO M+1
1256 B(A(I,1))=A(I,M+N+3)
1257 NEXT I
1260 PRINT "BASIC VARIABLES"
1270 FOR I=2 TO M+1
1280 PRINT "X(",A(I,1),")=";A(I,M+N+3)
1290 IF A(I,2)<>-1E14 THEN 1300
1295 IF A(I,M+N+3)>0 THEN 2690
1300 NEXT I
1310 PRINT "NON-BASIC VARIABLES"
1320 FOR J=1 TO M+N
1330 FOR I=2 TO M+1
1340 IF J=A(I,1) THEN 1370
1350 NEXT I
1360 PRINT "X(",J,")=";0
1370 NEXT J
1380 PRINT
1390 IF X=0 THEN 1420

```

```

1400 PRINT "MIN Z =" ; - A(M+2,M+N+3)
1410 GOTO 1430
1420 PRINT "MAX Z =" ; A(M+2,M+N+3)
1430 PRINT
1440 REM *****
1450 REM *          SENSITIVITY ANALYSIS          *
1460 REM *****
1470 PRINT "          ***** SENSITIVITY ANALYSIS *****"
1480 PRINT
1490 REM *****
1500 REM *          SHADOW PRICE CALCULATION          *
1510 REM *****
1520 PRINT "          *** SHADOW PRICE ***"
1530 PRINT "CONSTRAINT SHADOW-PRICE"
1540 PRINT "-----"
1550 FOR J=N+3 TO M+N+2
1560 IF A(1,J) = 0 THEN 1590
1570 PRINT J-N-2, A(M+2,J) - A(1,J)
1580 GOTO 1600
1590 PRINT J-N-2, - A(M+2,J)
1600 NEXT J
1610 PRINT
1620 REM *****
1630 REM *          RIGHT-HAND-SIDE-RANGE ANALYSIS          *
1640 REM *****
1650 PRINT "          *** RIGHT-HAND-SIDE-RANGE ***"
1660 DIM D(100), E(100)
1670 FOR J=N+3 TO M+N+2
1680 L1 = 1E+14
1690 FOR I=2 TO M+1
1700 IF A(I,J) <= 0 THEN 1740
1710 C(I) = A(1,M+N+3) / A(1,J)
1720 IF C(I) >= L1 THEN 1740
1730 L1 = C(I)
1740 NEXT I
1750 D(J-N-2) = L1
1760 L1 = 1E+14
1770 FOR I=2 TO M+1
1780 IF -A(I,J) <= 0 THEN 1820

```

```

1790 C(I) = -A(I,M+N+3)/A(I,J)
1800 IF C(I) >= L1 THEN 1820
1810 L1 = C(I)
1820 NEXT I
1830 E(J-N-2) = L1
1840 NEXT J
1850 PRINT "CONS. ORI-VAL DECREASE INCREASE",
1860 PRINT " MINIMUM MAXIMUM"
1870 PRINT "-----",
1880 PRINT "-----"
1890 FOR I=J TO M
1900 IF D(I) = 1E+14 THEN 1940
1910 IF E(I) = 1E+14 THEN 1960
1920 PRINT I, " ", O(I), D(I), E(I), O(I) - D(I), O(I) + E(I)
1930 GOTO 1970
1940 PRINT I, " ", O(I), "NO LIMIT", E(I), "0", O(I) + E(I)
1950 GOTO 1970
1960 PRINT I, " ", O(I), D(I), "NO LIMIT", O(I) - D(I), "NO LIMIT"
1970 NEXT I
1980 PRINT
1990 REM * * * * *
2000 REM * OBJECTIVE-FUNCTION-RANGING ANALYSIS *
2010 REM * * * * *
2020 PRINT " * * * OBJECTIVE FUNCTION RANGING * * *"
2030 FOR J=1 TO W
2040 D(J) = 1E+14
2050 E(J) = 1E+14
2060 NEXT J
2070 PRINT "D-VAR ORI-VAL DECREASE INCREASE",
2080 PRINT " MINIMUM MAXIMUM"
2090 PRINT "-----",
2100 PRINT "-----"
2110 IF X = 0 THEN 2360
2120 FOR J=1 TO W
2130 IF B(J) <> 0 THEN 2160
2140 D(J) = -A(M+2, J+2)
2150 GOTO 2310
2160 FOR I=2 TO M+1
2170 IF B(J) <> A(I,M+N+3) THEN 2300

```

```

2180 FOR R=3 TO M+N+2
2190 IF A(M+2,R) = 0 THEN 2280
2200 IF A(I,R) = 0 THEN 2280
2210 L1=A(M+2,R)/A(I,R)
2220 IF L1<0 THEN 2260
2230 IF L1>= D(J) THEN 2280
2240 D(J) = L1
2250 GOTO 2280
2260 IF ABS(L1)>= E(J) THEN 2280
2270 E(J) = ABS(L1)
2280 NEXT R
2290 GOTO 2310
2300 NEXT I
2310 NEXT J
2320 FOR J=1 TO W
2330 A(1,J+2) = -A(1,J+2)
2340 NEXT J
2350 GOTO 2560
2360 FOR J=1 TO W
2370 IF B(J)<>0 THEN 2400
2380 E(J) = -A(M+2,J+2)
2390 GOTO 2550
2400 FOR I=2 TO M+1
2410 IF B(J)<>A(I,M+N+3) THEN 2540
2420 FOR R=3 TO M+N+2
2430 IF A(M+2,R) = 0 THEN 2520
2440 IF A(I,R) = 0 THEN 2520
2450 L1=A(M+2,R)/A(I,R)
2460 IF L1<0 THEN 2500
2470 IF L1>= E(J) THEN 2520
2480 E(J) = L1
2490 GOTO 2520
2500 IF ABS(L1)>= D(J) THEN 2520
2510 D(J) = ABS(L1)
2520 NEXT R
2530 GOTO 2550
2540 NEXT I
2550 NEXT J
2560 FOR J=1 TO W

```

```

2570 IF D(J) = 1E+14 THEN 2620
2580 IF E(J) = 1E+14 THEN 2650
2590 PRINT "X(", J, ") ", A(1, J+2), D(J), E(J),
2600 PRINT A(1, J+2) - D(J), A(1, J+2) + E(J)
2610 GOTO 2670
2620 PRINT "X(", J, ") ", A(1, J+2), "NO LIMIT", E(J),
2630 PRINT "NO LIMIT", A(1, J+2) + E(J)
2640 GOTO 2670
2650 PRINT "X(", J, ") ", A(1, J+2), D(J), "NO LIMIT",
2660 PRINT A(1, J+2) - D(J), "NO LIMIT"
2670 NEXT J
2680 GOTO 2760
2690 PRINT
2700 PRINT "SINCE THE ARTIFICIAL VARIABLE X(",
2710 PRINT A(1, 1), ") DOES NOT EXIT FROM BASIS,"
2720 PRINT "A FEASIBLE SOLUTION DOES NOT EXIST"
2730 GOTO 2760
2740 PRINT "OBJECTIVE FUNCTION IS NOT BOUND",
2750 PRINT "BY CONSTRAINTS"
2760 PRINT "      *   END   *"
2770 END
3010 DATA 0,0,200,500,0,0,0,0
3020 DATA 3,0,0,1,1,0,0,40
3030 DATA 4,0,1.2,4,0,1,0,240
3040 DATA 5,0,.5,1,0,0,1,81

```

程序 LPSIM 中注释语句中文意思

```

0020 REM * 输入部分 *
0280 REM * 读入系数矩阵 *
0430 REM * 寻找 C - Z 和 Z 值 *
0560 REM * 循环开始 *
0600 REM * 寻找枢列 *
0690 REM * 确定是否到达最优 *
0760 REM * 寻找枢行 *
0890 REM * 枢运算 *
1030 REM * 新基变量进基 *
1080 REM * 寻找 C - Z 和 Z 值 *
1240 REM * 输出最优解和最优值 Z *
1450 REM * 敏感性分析 *
1500 REM * 影子价格计算 *

```


1630 REM * 右边常系数范围的分析 *

2000 REM * 目标函数中系数范围的分析 *

§1.2 线性规划问题及其数学模型

如何把有限的资源(例如机器、劳动人力、资金和原料)以最有效的方式分配给若干相互竞争的生产项目,使得成本最小或利润最大,这是经济管理工作经常遇到的问题。线性规划是优化这类问题的重要方法,也是管理科学中应用最广泛的一种决策技巧和工具。

线性规划模型的一般形式为

$$\max (\text{或} \min) Z = c_1 X_1 + c_2 X_2 + \cdots + c_w X_w$$

(目标函数)

$$\text{s. t.} \quad a_{11}x_1 + a_{12}x_2 + \cdots + a_{1w}x_w (*) b_1$$

(满足约束条件)

$$a_{21}x_1 + a_{22}x_2 + \cdots + a_{2w}x_w (*) b_2$$

$$\dots\dots\dots$$

$$a_{m1}x_1 + a_{m2}x_2 + \cdots + a_{mw}x_w (*) b_m$$

其中, x_j 表示第 j 个决策变量;

c_j 表示目标函数中 x_j 的系数;

a_{ij} 表示约束条件矩阵中的系数;

b_i 表示约束向量的系数;

(*)表示每个约束条件是 $\leq$, 或 $=$, 或 $\geq$;

m 是约束条件的个数, $i = 1, 2, \dots, m$;

w 是决策变量的个数, $j = 1, 2, \dots, w$ 。

在上述数学模型中, 目标函数是单一的、线性的。约束条件也假设为线性的。这是因为线性规划模型必须有两基本假设: 确定性假设和线性假设。

(1) 确定性假设

确定性假设要求所有的参数都为已知。在决策问题的领域里, 线性规划是一类确定性的决策模型。也就是说, 目标函数的系数 c_j , 约束条件的系数 a_{ij} , 右端系数 b_i , 都必须是常数。当这些系数不全时, 就要进行参数估计。决策者可以通过敏感性分析来考查系数变化的影响, 对那些估计的, 或随时间而变化的系数尤为重要。

(2) 线性假设

线性是某种数学函数性质, 它表明自变量与因变量之间的平稳关系, 其在图上可用直线表示。线性有三条具体性质: (1) 比例性; (2) 可加性; (3) 可分性。

比例性要求各种资源都与决策变量成比例, 而且必在变量的所有变化范围内都成立。例如在一个最大化目标函数 $200x_1 + 500x_2$ 中, 每个 x_1 单位对目标函数值提供200(比例量)。

可加性假设目标函数值或资源使用总量等于各决策变量提供的总和。如果上例的目标函数 $200x_1 + 500x_2$ 中, 设 $x_1 = 1$, $x_2 = 1$, 则可加性就认为 x_1 和 x_2 各自提供的量200和500的总和700就为目标函数的值。

可分性即指决策变量是可以连续变化的。也就是决策变量可以是分数或小数。